
Parameter Gathering Techniques for Data Driven Programming 
Renato G. Villacorte, Fairbank, Maslin, Maullin & Associates, Santa Monica, CA 

ABSTRACT 
Data Driven Programming maximizes procedural efficiency and lightens the workload of routine processing.  This 
paper presents techniques for gathering the parameters such as filenames and procedure results that can be used in 
subsequent routines and macro programming.  Techniques include utilizing the SYSEXEC, OUTPUT=, and ODS 
OUTPUT statements.  Examples of these coding techniques include how to read an MS Windows folder’s filenames 
into a SAS dataset, how to extract Proc results into datasets, and how to make SAS write programs based on these 
results.  

INTRODUCTION 
Macros greatly increase the efficiency of routine programming.  However, most macro programming depends upon a 
set of circumstances remaining completely constant.  What if you wanted your program to be dynamic and execute 
differently depending on file name or statistical result changes?  You would need a method of finding the most current 
data and a way of programming it into usable SAS code.  This paper describes the techniques used by an import 
macro.  Needing only the name of the file folder, the macro is able to locate the file name of a data file and 
accompanying map file.  Using the information found in the map file, it creates a SAS program for importing the data 
file.  Additional examples for gathering parameter data are also discussed. 

SOURCES OF PARAMETER DATA 
Parameters consist of filenames, variable names, variable values, or even lists that macros use to execute a function 
or routine.  Macros become more efficient when the program can locate its own parameters.  For example, a data 
import macro might find new variables added to a file or even find the most recent data file within a Windows folder.  
These parameters can be found within SAS datasets, PROC results, and even with commonly used DOS commands.  
Once the routine locates the parameter, it gets stored into a macro variable or dataset and then resolved in 
subsequent programming. 

SOURCES WITHIN SAS 
Practically every piece of information contained in or about a SAS dataset can be manipulated into a macro variable.  
Information about variables in a dataset can be found by using a DATA step, a CONTENTS procedure, or other 
procedure.  By using the OUT= option in PROC CONTENTS, you can retrieve variable names, lengths, formats, and 
even position within a dataset.  Using a DATA step with the SYMPUT statement you can retrieve an individual 
record’s values within a variable.  If you needed results from a statistical or analytical procedure like FREQ or SORT 
you could use a combination of the OUT= and SYMPUT techniques. 

EXTERNAL SOURCES 
Very often, the data that we wish to retrieve as a parameter is not yet in a SAS object.  The %SYSEXEC statement is 
useful in executing command line instructions.  This is done within a DOS session on the Windows XP™ platform.  
This gives you the flexibility and ability to manipulate data files within your operating system and set objects up for 
import into SAS.  Once imported into SAS, you can use the techniques mentioned above to store it as a parameter. 

EXAMPLES OF GATHERING PARAMETER DATA 
The following examples are portions of code that can appear within a data driven macro program.  The term “data 
driven” describes programs that react and change themselves according to the data they receive as they are 
executing.  For instance, an import macro might have to be told what folder to look in, but after knowing that it can 
determine a) the most recent or refreshed data file in the folder, b) the name of the dictionary file with the variable 
mapping information, and c) write its own import routine. 

EXAMPLE USING %SYSEXEC TO GET FILE NAMES AND IMPORT INTO SAS 
In our main example, the import macro is told to look in a folder named F:\Survey for a data file with a .dat extension 
and a map file with a .map extension.  The current goal is to get the actual names of the files within the Survey folder 
and then load it into a macro variable that can be used in the rest of the program.  The following resolved macro code 
uses the %SYSEXEC command to go to the specified folder and implement DIR commands. 

 
OPTIONS noxwait; 
%SYSEXEC CD F:\Survey; 
%SYSEXEC DIR *.dat > F:\Survey\datname.txt /b; 
%SYSEXEC DIR *.map > F:\Survey\mapname.txt /b; 

This code creates two files, datname.txt and mapname.txt, that contain the names of the files within the folder that 
have the respective extensions.  By default, when SYSEXEC is invoked, it opens a DOS command window that stays 
open even after executing the commands.  The noxwait option instructs SAS to close the window and continue 
executing the program.  Since we now know the names of the .txt files that contain the data we need, it is a simple 
matter to import that data into a SAS dataset. 

 
PROC IMPORT OUT=DATNAME 

DATAFILE= "F:\Survey\datname.txt"  DBMS=DLM REPLACE;   
DELIMITER='00'x;  GETNAMES=NO;  DATAROW=1; 

PROC IMPORT OUT=MAPNAME 
DATAFILE= "F:\Survey\mapname.txt"  DBMS=DLM REPLACE;  
DELIMITER='00'x;  GETNAMES=NO;  DATAROW=1; 

RUN; 
 

Now that we have the information we want in a SAS dataset, we can use the techniques that apply to any source 
within SAS. 

EXAMPLE OF USING DATA STEP AND SYMPUT TO DEFINE A MACRO VARIABLE 
In order for our overall import macro to work, we need a macro variable that represents the name of the file we want 
to import.  The end result of the preceding example code is a SAS dataset called DATNAME that contains the 
filename we are interested in.  Since the data we are looking for is stored as a record within a dataset, we can use the 
DATA _NULL_ method and SYMPUT call to define the macro variable. 

 
DATA _NULL_;  SET DATNAME; 

NAME= SUBSTR(VAR1, 1, (LENGTH(VAR1)-4));  
CALL SYMPUT ("DATFILE", NAME);  

RUN; 
 

The NAME variable is created by looking at a portion of VAR1 (the default variable name from the PROC IMPORT in 
the previous example).  The LENGTH statement in the SUBSTR function calculates the length of the filename after 
excluding the last 4 characters (the file extension .dat).  The SYMPUT function assigns the value of NAME into the 
macro variable named DATFILE.  Now, the overall macro can use &DATFILE whenever it needs to resolve the name 
of the data file being imported.  With this code technique, you can store any value found in a SAS dataset into a 
macro variable.  The only remaining challenge is to find ways to get the values you want into a record within a SAS 
dataset. 

EXAMPLES OF CREATING SAS DATASETS FROM PROCEDURES 
Most PROC’s are able to output some portion of their results with an OUT= our OUTPUT= option.   

 
PROC FREQ DATA=SURVEYDATA; 

TABLE GENDER /OUT=GENDERCOUNTS; 
RUN; 
 

This useful option in PROC FREQ creates a dataset called GENDERCOUNTS that contains the variables GENDER, 
COUNT, and PERCENT.  There is one record for every value of GENDER found in the SURVEYDATA dataset.  
Using the SYMPUT technique, you could store any or all of these values into a macro variable.  Perhaps, this could 
be useful in creating an annotate dataset for a subsequent graph. 

 
PROC CONTENTS DATA=SURVEYDATA 

OUT=VARIABLESPECS; 
RUN; 
 

This same option in PROC CONTENTS creates the VARIABLESPECS dataset with 40 variables describing the 
characteristics of each variable in SURVEYDATA.  These characteristics include variable name, label, type, length, 
format, and even column position within the dataset.  Getting these values into a macro variable could prove very 
useful for a wide variety of macro applications. 

EXAMPLES OF EXTRACTING DATASETS FROM THE OUTPUT DELIVERY SYSTEM (ODS) 
Each table that you see in the output window after running a FREQ, or MEANS, or LOGISTIC procedure originated as 
an ODS object.  The procedure calculates results and stores them as a table before displaying the information in the 
format dictated by the ODS.  To determine the names of these objects, you can run the PROC within ODS TRACE 
ON/OFF statements. 

ODS TRACE ON; 
PROC LOGISTIC DATA=SURVEYDATA; 
 MODEL Q2=Q1 Ver; 
RUN; 
ODS TRACE OFF; 
 

If we were interested in the table exhibiting the parameter estimates of this model, we would focus our attention on 
the portion of the log stating: 

 
Output Added: 
------------- 
Name:       ParameterEstimates 
Label:      Parameter Estimates 
Template:   Stat.Logistic.ParameterEstimates 
Path:       Logistic.ParameterEstimates 
------------- 
 

The trace tells us that ODS names the table we are interested in as ParameterEstimates.  In order to save this object 
as a SAS dataset, we would need to use the ODS OUTPUT statement. 

 
ODS OUTPUT 
 PARAMETERESTIMATES(MATCH_ALL=DATASETNAMES)=PARAMSET; 
PROC LOGISTIC DATA=SURVEYDATA; 
 MODEL Q2=Q1 Ver; 
RUN; 
ODS OUTPUT CLOSE; 
 

The OUTPUT statement instructs the ODS system to save the PARAMETERESTIMATES object as a SAS dataset 
called PARAMSET.  Now that the parameter information is in a dataset, you can use the SYMPUT technique to load 
information into your macro variables.  This can be extremely useful when experimenting with a number of different 
logistic models and you want to store your Chi-square probability values in a dataset.  All in all, if ODS creates it, it 
can be saved in a dataset. 

UTILIZING PARAMETER DATA 
MACRO VARIABLES 
Once assigned to a macro variable, the parameter information can be easily inserted into subsequent programming. 

 
PROC IMPORT OUT=NEWDATASET 

DATAFILE= "F:\Survey\&DATFILE..DAT"  DBMS=DLM REPLACE;   
DELIMITER='00'x;  GETNAMES=YES;  DATAROW=2; 

RUN; 
 

The macro variable DATFILE is resolved within this import procedure by surrounding the variable name with an 
ampersand and period.  Note that double quotes need to be used here so the macro variable will be resolved as it is 
read by the macro processor.  A more comprehensive tutorial on resolving macro variables can be found in 
Carpenter’s Complete Guide to the SAS Macro Language. 

DATA DRIVEN PROGRAMMING 
A more sophisticated use of these techniques is to create programs based on the parameters you gathered.  In our 
first example, the macro found the name of a data file and a map file.  Let’s say the data file was a fixed column ASCII 
file and the map contained the column names and starting locations for the data.  In this case, the map file is in a 
standard format so that we can easily pre-program to import it.  So now, we have dataset containing a column for the 
variable name, starting column, and width. 

MAPFILE DATASET: 
NAME       COL  WIDTH  DUMMY 
VER          7    1      1                               
Q1           8    1      1                               
Q2           9    1      1                               
Q3          10    1      1                               
Q4          11    1      1                               
AGE         12    2      1                               
SEX         36    1      1                               
PARTY       37    1      1                               
ZIP         38    5      1                               
PRECINCT    43    8      1                               
 

Collectively, this dataset is the parameter information that we need to program an import routine for the data file.  We 
can use a DATA _NULL_ step to create a new file containing SAS code based on the MAPFILE dataset. 

 
DATA _NULL_;  SET MAPFILE; 
BY DUMMY;  
FILE "F:\Survey\READ_DATAFILE.SAS"; 
 

When importing the map file, we created a dummy variable equal to 1 for all rows.  This allows us to use a BY 
statement to mark the beginning and ending of the dataset.  At the beginning or our program, we want to print some 
remarks and some initial statements that are common to all of our projects.  Note that the data file name that we 
obtained earlier in our example is resolved in the FILENAME statement. 

 
IF FIRST.DUMMY THEN DO;  
PUT "* PROGRAM FOR READING PROJECT Survey;"; 
PUT "FILENAME DATAFILE ='F:\Survey\&DATFILE..DAT';";  
PUT "LIBNAME DATAFOLDER  'F:\Survey';"; 
PUT "DATA DATAFOLDER.RAWDATA;"; 
PUT "INFILE DATAFILE MISSOVER LRECL=800;"; 
PUT "INPUT"; 
END;  
 

We now have the beginning of a DATA step that will import the data file with the INFILE and INPUT statements.  To 
input the specific variables, we will use the MAPFILE dataset variables to fill in the INPUT parameters. 

 
IF WIDTH <3 THEN DO;  

PUT "@" COL " " NAME " " WIDTH +(-1) ".0"; 
END;  
ELSE DO;  

PUT "@" COL " " NAME " $CHAR" WIDTH +(-1)  ".0"; 
END;  
IF LAST.DUMMY THEN PUT ";  RUN;";  
 

The preceding block of code treats variables differently depending on their column width.  By default, any column that 
is 2 or less digits wide will be treated as a numeric variable.  All others will be read in as character strings.  Note that 
put statements contain a combination of quoted text and variables contained in the MAPFILE dataset.  However, after 
each variable, SAS likes to move the cursor over a space.  We add the +(-1) to tell the cursor to move back and place 
the subsequent text immediately after the variable value.  Finally, when we reach the end of the dataset, as indicated 
with LAST.DUMMY, we close off the INPUT statement with a semi-colon and finish off the block of code with a RUN 
statement.  Now, if we look in the newly created F:\Survey\READ_DATAFILE.SAS program we find: 

* PROGRAM FOR READING PROJECT Survey; 
FILENAME DATAFILE = 'F:\Survey\SurveyData.DAT'; 
LIBNAME DATAFOLDER  'F:\Survey'; 
DATA DATAFOLDER.RAWDATA; 
INFILE DATAFILE MISSOVER LRECL=800; 
INPUT 
   @7    VER   1.0 
   @8    Q1    1.0 
   @9    Q2    1.0 
   @10   Q3    1.0 
   @11   Q4    1.0 
   @12   AGE   2.0 
   @36   SEX   1.0 
   @37   PARTY 1.0 
   @38   ZIP   $CHAR5.0 
   @43   PRECINCT  $CHAR8.0 
;  RUN; 
 

We can run this program from the original macro with a %INCLUDE statement or leave it as a stand-alone program.  
After running, we now have a dataset called RAWDATA in the DATAFOLDER library. 

CONCLUSION  
Using SAS Macros and parameter gathering techniques, the preceding program was written by only providing the 
folder name of the project.  The macros looked in the folder, determined the file names, and created the import 
program based on a map file.  These data driven macros are extremely powerful when your programming projects 
follow a common routine.  These examples are based on macros used to read in survey data that comes in a 
common format.  The only differences between projects are the names of the files (not the extensions) and the 
variable maps.  This program has saved us hundreds of programming hours as we process dozens of surveys every 
month and do not have to program each individually.  Hopefully, you can apply these techniques to your routines and 
realize the same benefit. 

